Задача 8. Дан список, в каждой строке которого указано, какая страна какой стране сколько миллиардов должна. Оптимизировать список долгов, используя взаиморасчеты. 

Пример входного файла: 

England France 2

USA England 5

France USA 8

Пример выходного файла: 

France England 3

France USA 3

Решение. Следует просматривать список долгов, и находить всевозможные пары, которые можно «склеить», то если кредитор одной пары совпадает с должником другой пары. Если удаётся обнаружить две такие пары, то возможны две ситуации:

1. Страна А должна стране Б, и страна Б должна стране А.

2. Страна А должна стране Б, и страна Б должна стране С.

В персом случае (Страна А должна стране Б, и страна Б должна стране А) следует найти минимальную сумму и вычесть её из максимальной, элемент списка долгов с минимальной суммой удалить из списка. 

Для удаления строки из списка долгов не обязательно физически удалять эту строку, мучаясь потом со сдвигом остальных элементов списка. Для удаления элемента списка долгов мы введём множество индексов, в котором в начале процесса оптимизации будут все индексы от первого до последнего. Удалять строку списка долгов будем просто удаляя элемент множества индексов. Но впредь работать только с теми строками списка долгов, индексы которых есть в этом множестве, и выводить в качестве ответа список оптимизированных долгов только с индексами из этого множества. 
Путь теперь страна А должна стране Б х млрд., и страна Б должна стране С y млрд.

Возможны три случая:

– x=y; 

– x<y;

– x>y.

Если x=y, то ясно, что долг страны А можно непосредственно перенаправить в страну С, а долг страны Б удалить из списка.

Если x<y, то эти х млрд. направляем из страны А сразу в страну С, и тогда страна Б будет должна стране С y–x млрд.

Если x>y, то эти х–y млрд. направляем из страны А сразу в страну С, а y млрд. направляем из страны А в страну Б. 
Просматриваем так список всех долгов до тех пор, пока не будет сделано ни одной оптимизации. Для этого будем использовать флаг, логическую переменную, которой перед очередным просмотром будем присваивать истину, а в случае, если сделана хотя бы одна оптимизация, будем присваивать ложь.

После этого список долгов будет оптимизирован. Далее следует исходный код программы на языке Pascal с подробными комментариями.

program q;

  uses crt;

  const n=255; {Максимальное количество строк списка}

  type String20=string[20]; {Тип для хранения одной записи о долге}

       Dolg=record

         state1,state2:string20;

         money:integer

       end;

       TClub=array [1..n] of Dolg; 
         {Тип для хранения всей системы долгов}

       SetByte=Set of Byte; 
{Множество номеров строк, которые не стёрты}

  var f,g:Text; {Переменные-файлы для входных и выходных данных}

      x:TClub; {Массив для хранения записей о долгах}

      h,hh:string; {Строки для обработки входных данных}

      p1,p2,p, 

         {переменные для поиска пробелов в строках входных файлов}

      s,code, {переменные для перевода строковых данных в числовые}

      k, {Количество строк во входном файле}

      i,j:integer; {Счётчики}

      exist:SetByte; {Множество для хранения одной записи о долге}

      L:Boolean; 
        {Флаг-признак того, что больше в списке оптимизировать нечего}

  Function card(y:SetByte):integer; {Количество элементов множества}

    var i,s:integer;

    begin

      s:=0; {счетчик элементов множества}

      for i:=1 to 255 do if (i in y) then inc(s);

      card:=s

    end;

begin

  TextBackGround(1); TextColor(14); ClrScr; {Настройка цвета экрана}

  Assign(f,'c:/par.in'); reset(f); {Готовы читать файл}

  k:=0;{Количество долгов}

  while not(eof(f)) do {Пока не конец файла}

    begin

      readln(f,h); p1:=1; hh:='';inc(k);

      while h[p1]<>' ' do inc(p1); {Нашли первый пробел}

      for p:=1 to p1-1 do hh:=hh+h[p]; {Получили первую страну}

      x[k].State1:=hh; p2:=p1+1; hh:='';

      while h[p2]<>' ' do inc(p2); {Ищем второй пробел}

      for p:=p1+1 to p2-1 do hh:=hh+h[p]; {Строим название второй страны}

      x[k].State2:=hh; hh:='';

      for p:=p2+1 to length(h) do hh:=hh+h[p]; {Строим строку из суммы долга}

      val(hh,s,code); x[k].money:=s; {Определили долг}

    end;

  close(f); {Данные прочитаны и занесены в память

            k=количество записей о долгах}

  for i:=1 to k do {Выводим прочитанные данные}

    writeln(x[i].State1,' ',x[i].State2,' ',x[i].money);

  exist:=[1..k]; {Множество индексов рассматриваемых долгов}

  {По ходу взаимозачета долгов некоторые долги будут

   удалятся, и из множества рассматриваемых долгов

   будут удалятся их индексы с тем чтобы эти долги больше не   

   рассматривать и не выдавать в ответ}

  Repeat

    L:=True; {Флаг, признак того, что список долгов оптимизирован}

    for i:=1 to k do

    for j:=1 to k do

      begin

        if (i<>j)and(i in Exist)and(j in Exist)

          then {перебираем всевозможные пары долгов}

            begin

            {Сравниваем, можно ли построить ветку дерева x[i] -> x[j]}

            if (x[i].State2=x[j].state1)and(i in Exist)and(j in exist)

                then

                  begin

                    {Получилась ветка. Оптимизируем ветку}

                    if x[j].state2=x[i].state1 
{Если ветка зациклилась:

                         A должен В, а В должен А}

                      then

                        begin

                          if (x[i].money=x[j].money)and(i in Exist)and(j in exist)

                            then

                              begin

                                {Удаляем оба долга из списка долгов. Полный взаимозачет}

                                 exist:=exist-[i]; exist:=exist-[j]; L:=False

                              end;

                          if (x[j].money<x[i].money)and(i in Exist)and(j in exist)

                            then

                              begin

                                x[i].money:=x[i].money-x[j].money;

                                Exist:=Exist-[j]; L:=False

                              end

                        end

                      else

                        begin {Ветка не цикличная}

                          if (x[i].money=x[j].money)and(i in Exist)and(j in exist)

                            then

                              begin

                                x[i].state2:=x[j].state2; 

      {перенаправили долг целиком в другую страну}

  {Теперь j-й долг следует удалить}

                                Exist:=Exist-[j]; L:=False

                              end

                            else

                        if (x[i].money<x[j].money)and(i in Exist)and(j in exist)

                         {Проверку на наличие номеров долгов во множестве

                         Exist следует делать, потому что уже не нужно рассматривать

                         строки, которые только что были удалены

                         внутри этого шага цикла}

                                 then

                                   begin

                                     x[i].State2:=x[j].state2;                              
x[j].money:=x[j].money-x[i].money; L:=False

                                    end

                                  else

                                    if (i in Exist)and(j in exist)

                                      then

                                        begin

                                    x[i].money:=x[i].money-x[j].money;

                                          x[j].State1:=x[i].state1; 
   L:=False;                          
                                        end

                        end

                  end

            end

      end

    Until L;

    writeln; TextColor(11); assign(g,'c:/par.out'); rewrite(g);

    {Готовы писать в выходной файл}

    if card(exist)=0 {Проверка оптимизированного списка на пустоту}

      then

        begin

          writeln('Все долги взаимозачтены');

          writeln(g,'Все долги взаимозачтены')

        end

      else

        begin

          for i:=1 to k do

            if (i in Exist) {Выводим только невыброшенные строки}

              then

                begin

                  writeln(x[i].State1,' ',x[i].State2,' ',x[i].money);

                writeln(g,x[i].State1,' ',x[i].State2,' ',x[i].money);

                end

        end;

  close(g);{Закрыли готовый файл с ответами}

  repeat until keypressed;

end.
